

[Description](#)  
[Options](#)  
[References](#)[Quick start](#)  
[Remarks and examples](#)  
[Also see](#)[Menu](#)  
[Stored results](#)[Syntax](#)  
[Methods and formulas](#)

## Description

`mprobit` fits a multinomial probit (MNP) model for a categorical dependent variable with outcomes that have no natural ordering. The actual values taken by the dependent variable are irrelevant. The error terms are assumed to be independent, standard normal, random variables. `cmmprobit` relaxes the independence of irrelevant alternatives assumption by specifying correlated latent-variable errors. `cmmprobit` also allows heteroskedastic latent-variable errors and alternative-specific independent variables.

## Quick start

Multinomial probit model of `y` on `x1`, `x2`, and categorical `a`

```
mprobit y x1 x2 i.a
```

Same as above, but use `a` as the base outcome `y = 3`

```
mprobit y x1 x2 i.a, baseoutcome(3)
```

Probit variance parameterization of differenced latent errors

```
mprobit y x1 x2 i.a, probitparam
```

Multiple-imputation estimates with Monte Carlo errors from `mi` set data

```
mi estimate, merror: mprobit y x1 x2 i.a
```

## Menu

Statistics > Categorical outcomes > Multinomial probit regression

Syntax

```
mprobit depvar [indepvars] [if] [in] [weight] [, options]
```

<i>options</i>	Description
Model	
<code>noconstant</code>	suppress constant terms
<code>baseoutcome(# <i>lbl</i>)</code>	outcome used to normalize location
<code>probitparam</code>	use the probit variance parameterization
<code>constraints(<i>constraints</i>)</code>	apply specified linear constraints
SE/Robust	
<code>vce(<i>vcetype</i>)</code>	<i>vcetype</i> may be <code>oim</code> , <code>robust</code> , <code>cluster <i>clustvar</i></code> , <code>opg</code> , <code>bootstrap</code> , or <code>jackknife</code>
Reporting	
<code>level(#)</code>	set confidence level; default is <code>level(95)</code>
<code>nocnsreport</code>	do not display constraints
<code>display_options</code>	control columns and column formats, row spacing, line width, display of omitted variables and base and empty cells, and factor-variable labeling
Integration	
<code>intpoints(#)</code>	number of quadrature points
Maximization	
<code>maximize_options</code>	control the maximization process; seldom used
<code>collinear</code>	keep collinear variables
<code>coeflegend</code>	display legend instead of statistics

*indepvars* may contain factor variables; see [U] 11.4.3 Factor variables.

`bayes`, `bayesboot`, `bootstrap`, `by`, `collect`, `fp`, `jackknife`, `mi estimate`, `rolling`, `statsby`, and `svy` are allowed; see [U] 11.1.10 Prefix commands. For more details, see [BAYES] bayes: mprobit.

`vce(bootstrap)` and `vce(jackknife)` are not allowed with the `mi estimate` prefix; see [MI] mi estimate.

Weights are not allowed with the `bootstrap` prefix; see [R] bootstrap.

`vce()` and weights are not allowed with the `svy` prefix; see [SVY] svy.

`fweights`, `iweights`, and `pweights` are allowed; see [U] 11.1.6 weight.

`collinear` and `coeflegend` do not appear in the dialog box.

See [U] 20 Estimation and postestimation commands for more capabilities of estimation commands.

Options

Model
<code>noconstant</code> suppresses the $J - 1$ constant terms.
<code>baseoutcome(# <i>lbl</i>)</code> specifies the outcome used to normalize the location of the latent variable. The base outcome may be specified as a number or a label. The default is to use the most frequent outcome. The coefficients associated with the base outcome are zero.

`probitparam` specifies to use the probit variance parameterization by fixing the variance of the differenced latent errors between the scale and the base alternatives to be one. The default is to make the variance of the base and scale latent errors one, thereby making the variance of the difference to be two.

`constraints`(*constraints*); see [R] [Estimation options](#).

#### SE/Robust

`vce`(*vcetype*) specifies the type of standard error reported, which includes types that are derived from asymptotic theory (`oim`, `opg`), that are robust to some kinds of misspecification (`robust`), that allow for intragroup correlation (`cluster` *clustvar*), and that use bootstrap or jackknife methods (`bootstrap`, `jackknife`); see [R] [vce\\_option](#).

If specifying `vce(bootstrap)` or `vce(jackknife)`, you must also specify `baseoutcome()`.

#### Reporting

`level(#)`, `nocnsreport`; see [R] [Estimation options](#).

`display_options`: `nocl`, `nopvalues`, `noomitted`, `vsquish`, `noemptycells`, `baselevels`, `allbaselevels`, `novlabel`, `fvwrap(#)`, `fvwrapon(style)`, `cformat(%fmt)`, `pformat(%fmt)`, `sformat(%fmt)`, and `no1stretch`; see [R] [Estimation options](#).

#### Integration

`intpoints(#)` specifies the number of Gaussian quadrature points to use in approximating the likelihood. The default is `intpoints(15)`.

#### Maximization

`maximize_options`: `difficult`, `technique`(*algorithm\_spec*), `iterate(#)`, `[no]log`, `trace`, `gradient`, `showstep`, `hessian`, `showtolerance`, `tolerance(#)`, `ltolerance(#)`, `nrtolerance(#)`, `nonrtolerance`, and `from(init_specs)`; see [R] [Maximize](#). These options are seldom used.

Setting the optimization type to `technique(bhhh)` resets the default `vcetype` to `vce(opg)`.

The following options are available with `mprobit` but are not shown in the dialog box:

`collinear`, `coeflegend`; see [R] [Estimation options](#).

## Remarks and examples

The MNP model is used with discrete dependent variables that take on more than two outcomes that do not have a natural ordering. The stochastic error terms for this implementation of the model are assumed to have independent, standard normal distributions. To use `mprobit`, you must have one observation for each decision maker in the sample. See [CM] [cmmprobit](#) for another implementation of the MNP model that permits correlated and heteroskedastic errors and is suitable when you have data for each alternative that a decision maker faced.

The MNP model is frequently motivated using a latent-variable framework. The latent variable for the  $j$ th alternative,  $j = 1, \dots, J$ , is

$$\eta_{ij} = \mathbf{z}_i \boldsymbol{\alpha}_j + \xi_{ij}$$

where the  $1 \times q$  row vector  $\mathbf{z}_i$  contains the observed independent variables for the  $i$ th decision maker. Associated with  $\mathbf{z}_i$  are the  $J$  vectors of regression coefficients  $\boldsymbol{\alpha}_j$ . The  $\xi_{i,1}, \dots, \xi_{i,J}$  are distributed independently and identically standard normal. The decision maker chooses the alternative  $k$  such that  $\eta_{ik} \geq \eta_{im}$  for  $m \neq k$ .

Suppose that case  $i$  chooses alternative  $k$ , and take the difference between latent variable  $\eta_{ik}$  and the  $J - 1$  others:

$$\begin{aligned} v_{ijk} &= \eta_{ij} - \eta_{ik} \\ &= \mathbf{z}_i(\boldsymbol{\alpha}_j - \boldsymbol{\alpha}_k) + \xi_{ij} - \xi_{ik} \\ &= \mathbf{z}_i \boldsymbol{\gamma}_{j'} + \epsilon_{ij'} \end{aligned} \tag{1}$$

where  $j' = j$  if  $j < k$  and  $j' = j - 1$  if  $j > k$  so that  $j' = 1, \dots, J - 1$ .  $\text{Var}(\epsilon_{ij'}) = \text{Var}(\xi_{ij} - \xi_{ik}) = 2$  and  $\text{Cov}(\epsilon_{ij'}, \epsilon_{il'}) = 1$  for  $j' \neq l'$ . The probability that alternative  $k$  is chosen is

$$\begin{aligned} \Pr(i \text{ chooses } k) &= \Pr(v_{i1k} \leq 0, \dots, v_{i,J-1,k} \leq 0) \\ &= \Pr(\epsilon_{i1} \leq -\mathbf{z}_i \boldsymbol{\gamma}_1, \dots, \epsilon_{i,J-1} \leq -\mathbf{z}_i \boldsymbol{\gamma}_{J-1}) \end{aligned}$$

Hence, evaluating the likelihood function involves computing probabilities from the multivariate normal distribution. That all the covariances are equal simplifies the problem somewhat; see [Methods and formulas](#) for details.

In (1), not all  $J$  of the  $\boldsymbol{\alpha}_j$  are identifiable. To remove the indeterminacy,  $\boldsymbol{\alpha}_l$  is set to the zero vector, where  $l$  is the base outcome as specified in the `baseoutcome()` option. That fixes the  $l$ th latent variable to zero so that the remaining variables measure the attractiveness of the other alternatives relative to the base.

## ► Example 1

As discussed in [example 1](#) of [\[R\] mlogit](#), we have data on the type of health insurance available to 616 psychologically depressed subjects in the United States (Tarlov et al. 1989; Wells et al. 1989). Patients may have either an indemnity (fee-for-service) plan or a prepaid plan such as an HMO, or the patient may be uninsured. Demographic variables include age, gender, race, and site. Indemnity insurance is the most popular alternative, so `mprobit` will choose it as the base outcome by default.

```
. use https://www.stata-press.com/data/r19/sysdsn1
(Health insurance data)

. mprobit insure age male nonwhite i.site

Iteration 0:  Log likelihood = -535.89424
Iteration 1:  Log likelihood = -534.56173
Iteration 2:  Log likelihood = -534.52835
Iteration 3:  Log likelihood = -534.52833

Multinomial probit regression                                Number of obs =    615
                                                            Wald chi2(10) =   40.18
Log likelihood = -534.52833                                Prob > chi2   = 0.0000
```

insure	Coefficient	Std. err.	z	P> z	[95% conf. interval]	
Indemnity	(base outcome)					
Prepaid						
age	-.0098536	.0052688	-1.87	0.061	-.0201802	.000473
male	.4774678	.1718316	2.78	0.005	.1406841	.8142515
nonwhite	.8245003	.1977582	4.17	0.000	.4369013	1.212099
site						
2	.0973956	.1794546	0.54	0.587	-.2543289	.4491201
3	-.495892	.1904984	-2.60	0.009	-.869262	-.1225221
_cons	.22315	.2792424	0.80	0.424	-.324155	.7704549
Uninsure						
age	-.0050814	.0075327	-0.67	0.500	-.0198452	.0096823
male	.3332637	.2432986	1.37	0.171	-.1435929	.8101203
nonwhite	.2485859	.2767734	0.90	0.369	-.29388	.7910518
site						
2	-.6899485	.2804497	-2.46	0.014	-1.23962	-.1402771
3	-.1788447	.2479898	-0.72	0.471	-.6648957	.3072063
_cons	-.9855917	.3891873	-2.53	0.011	-1.748385	-.2227986



The likelihood function for `mprobit` is derived under the assumption that all decision-making units face the same choice set, which is the union of all outcomes observed in the dataset. If that is not true for your model, then an alternative is to use the `cmmprobit` command, which does not require this assumption. To do that, you will need to expand the dataset so that each decision maker has  $k_i$  observations, where  $k_i$  is the number of alternatives in the choice set faced by decision maker  $i$ . You will also need to create a binary variable to indicate the choice made by each decision maker. Moreover, you will need to use the `correlation(independent)` and `stddev(homoskedastic)` options with `cmmprobit` unless you have alternative-specific variables.

## Stored results

mprobit stores the following in `e()`:

### Scalars

<code>e(N)</code>	number of observations
<code>e(k_out)</code>	number of outcomes
<code>e(k_points)</code>	number of quadrature points
<code>e(k)</code>	number of parameters
<code>e(k_eq)</code>	number of equations in <code>e(b)</code>
<code>e(k_eq_model)</code>	number of equations in overall model test
<code>e(k_indvars)</code>	number of independent variables
<code>e(k_dv)</code>	number of dependent variables
<code>e(df_m)</code>	model degrees of freedom
<code>e(ll)</code>	log likelihood
<code>e(N_clust)</code>	number of clusters
<code>e(chi2)</code>	$\chi^2$
<code>e(p)</code>	$p$ -value for model test
<code>e(k_eq_base)</code>	equation number of the base outcome
<code>e(baseout)</code>	the value of <code>depvar</code> to be treated as the base outcome
<code>e(ibaseout)</code>	index of the base outcome
<code>e(const)</code>	0 if <code>noconstant</code> is specified, 1 otherwise
<code>e(probitparam)</code>	1 if <code>probitparam</code> is specified, 0 otherwise
<code>e(rank)</code>	rank of <code>e(V)</code>
<code>e(ic)</code>	number of iterations
<code>e(rc)</code>	return code
<code>e(converged)</code>	1 if converged, 0 otherwise

### Macros

<code>e(cmd)</code>	mprobit
<code>e(cmdline)</code>	command as typed
<code>e(depvar)</code>	name of dependent variable
<code>e(indvars)</code>	independent variables
<code>e(wtype)</code>	weight type
<code>e(wexp)</code>	weight expression
<code>e(title)</code>	title in estimation output
<code>e(clustvar)</code>	name of cluster variable
<code>e(chi2type)</code>	Wald, type of model $\chi^2$ test
<code>e(vce)</code>	<code>vcetype</code> specified in <code>vce()</code>
<code>e(vcetype)</code>	title used to label Std. err.
<code>e(outeqs)</code>	outcome equations
<code>e(out#)</code>	outcome labels, <code># = 1, \dots, e(k_out)</code>
<code>e(opt)</code>	type of optimization
<code>e(which)</code>	max or min; whether optimizer is to perform maximization or minimization
<code>e(ml_method)</code>	type of ml method
<code>e(user)</code>	name of likelihood-evaluator program
<code>e(technique)</code>	maximization technique
<code>e(properties)</code>	b V
<code>e(predict)</code>	program used to implement <code>predict</code>
<code>e(marginsnotok)</code>	predictions disallowed by <code>margins</code>
<code>e(marginsdefault)</code>	default <code>predict()</code> specification for <code>margins</code>
<code>e(asbalanced)</code>	factor variables <code>fvset</code> as <code>asbalanced</code>
<code>e(asobserved)</code>	factor variables <code>fvset</code> as <code>asobserved</code>

Matrices	
e(b)	coefficient vector
e(outcomes)	outcome values
e(Cns)	constraints matrix
e(ilog)	iteration log (up to 20 iterations)
e(gradient)	gradient vector
e(V)	variance–covariance matrix of the estimators
e(V_modelbased)	model-based variance
Functions	
e(sample)	marks estimation sample

In addition to the above, the following is stored in `r()`:

Matrices	
r(table)	matrix containing the coefficients with their standard errors, test statistics, $p$ -values, and confidence intervals

Note that results stored in `r()` are updated when the command is replayed and will be replaced when any `r-class` command is run after the estimation command.

## Methods and formulas

See [Cameron and Trivedi \(2005, chap. 15\)](#) for a discussion of multinomial models, including multinomial probit. [Long and Freese \(2014, chap. 8\)](#) discuss the multinomial logistic, multinomial probit, and stereotype logistic regression models, with examples using Stata.

As discussed in [Remarks and examples](#), the latent variables for a  $J$ -alternative model are  $\eta_{ij} = \mathbf{z}_i \boldsymbol{\alpha}_j + \xi_{ij}$ , for  $j = 1, \dots, J$ ,  $i = 1, \dots, n$ , and  $\{\xi_{i,1}, \dots, \xi_{i,J}\} \sim \text{i.i.d.} N(0, 1)$ . The experimenter observes alternative  $k$  for the  $i$ th observation if  $\eta_{ik} > \eta_{il}$  for  $l \neq k$ . For  $j \neq k$ , let

$$\begin{aligned} v_{ij'} &= \eta_{ij} - \eta_{ik} \\ &= \mathbf{z}_i (\boldsymbol{\alpha}_j - \boldsymbol{\alpha}_k) + \xi_{ij} - \xi_{ik} \\ &= \mathbf{z}_i \boldsymbol{\gamma}_{j'} + \epsilon_{ij'} \end{aligned}$$

where  $j' = j$  if  $j < k$  and  $j' = j - 1$  if  $j > k$  so that  $j' = 1, \dots, J - 1$ .  $\boldsymbol{\epsilon}_i = (\epsilon_{i1}, \dots, \epsilon_{i,J-1}) \sim MVN(\mathbf{0}, \boldsymbol{\Sigma})$ , where

$$\boldsymbol{\Sigma} = \begin{pmatrix} 2 & 1 & 1 & \dots & 1 \\ 1 & 2 & 1 & \dots & 1 \\ 1 & 1 & 2 & \dots & 1 \\ \vdots & \vdots & \vdots & \ddots & \vdots \\ 1 & 1 & 1 & \dots & 2 \end{pmatrix}$$

Denote the deterministic part of the model as  $\lambda_{ij'} = \mathbf{z}_i \boldsymbol{\gamma}_{j'}$ ; the probability that subject  $i$  chooses outcome  $k$  is

$$\begin{aligned} \Pr(y_i = k) &= \Pr(v_{i1} \leq 0, \dots, v_{i,J-1} \leq 0) \\ &= \Pr(\epsilon_{i1} \leq -\lambda_{i1}, \dots, \epsilon_{i,J-1} \leq -\lambda_{i,J-1}) \\ &= \frac{1}{(2\pi)^{(J-1)/2} |\boldsymbol{\Sigma}|^{1/2}} \int_{-\infty}^{-\lambda_{i1}} \dots \int_{-\infty}^{-\lambda_{i,J-1}} \exp\left(-\frac{1}{2} \mathbf{z}' \boldsymbol{\Sigma}^{-1} \mathbf{z}\right) d\mathbf{z} \end{aligned}$$

Because of the exchangeable correlation structure of  $\Sigma$  ( $\rho_{ij} = 1/2$  for all  $i \neq j$ ), we can use Dunnett's (1989) result to reduce the multidimensional integral to one dimension:

$$\Pr(y_i = k) = \frac{1}{\sqrt{\pi}} \int_0^\infty \left\{ \prod_{j=1}^{J-1} \Phi(-z\sqrt{2} - \lambda_{ij}) + \prod_{j=1}^{J-1} \Phi(z\sqrt{2} - \lambda_{ij}) \right\} e^{-z^2} dz$$

Gaussian quadrature is used to approximate this integral, resulting in the  $K$ -point quadrature formula

$$\Pr(y_i = k) \approx \frac{1}{2} \sum_{k=1}^K w_k \left\{ \prod_{j=1}^{J-1} \Phi(-\sqrt{2x_k} - \lambda_{ij}) + \prod_{j=1}^{J-1} \Phi(\sqrt{2x_k} - \lambda_{ij}) \right\}$$

where  $w_k$  and  $x_k$  are the weights and roots of the Laguerre polynomial of order  $K$ . In `mprobit`,  $K$  is specified by the `intpoints()` option.

This command supports the Huber/White/sandwich estimator of the variance and its clustered version using `vce(robust)` and `vce(cluster clustvar)`, respectively. See [P] [\\_robust](#), particularly [Maximum likelihood estimators](#) and [Methods and formulas](#).

`mprobit` also supports estimation with survey data. For details on VCEs with survey data, see [SVY] [Variance estimation](#).

## References

- Cameron, A. C., and P. K. Trivedi. 2005. *Microeconometrics: Methods and Applications*. New York: Cambridge University Press.
- Dunnett, C. W. 1989. Algorithm AS 251: Multivariate normal probability integrals with product correlation structure. *Journal of the Royal Statistical Society, C ser.*, 38: 564–579. <https://doi.org/10.2307/2347754>.
- Haan, P., and A. Uhlenborff. 2006. Estimation of multinomial logit models with unobserved heterogeneity using maximum simulated likelihood. *Stata Journal* 6: 229–245.
- Hole, A. R. 2007. Fitting mixed logit models by using maximum simulated likelihood. *Stata Journal* 7: 388–401.
- Long, J. S., and J. Freese. 2014. *Regression Models for Categorical Dependent Variables Using Stata*. 3rd ed. College Station, TX: Stata Press.
- Tarlov, A. R., J. E. Ware, Jr., S. Greenfield, E. C. Nelson, E. Perrin, and M. Zubkoff. 1989. The medical outcomes study. An application of methods for monitoring the results of medical care. *Journal of the American Medical Association* 262: 925–930. <https://doi.org/10.1001/jama.1989.03430070073033>.
- Wells, K. B., R. D. Hays, M. A. Burnam, W. H. Rogers, S. Greenfield, and J. E. Ware, Jr. 1989. Detection of depressive disorder for patients receiving prepaid or fee-for-service care. Results from the Medical Outcomes Survey. *Journal of the American Medical Association* 262: 3298–3302. <https://doi.org/10.1001/jama.1989.03430230083030>.



## Also see

- [R] **mprobit postestimation** — Postestimation tools for mprobit
- [R] **clogit** — Conditional (fixed-effects) logistic regression
- [R] **mlogit** — Multinomial (polytomous) logistic regression
- [R] **ologit** — Ordered logistic regression
- [R] **oprobit** — Ordered probit regression
- [BAYES] **bayes: mprobit** — Bayesian multinomial probit regression
- [CM] **cmmprobit** — Multinomial probit choice model
- [CM] **nlogit** — Nested logit regression
- [MI] **Estimation** — Estimation commands for use with mi estimate
- [SVY] **svy estimation** — Estimation commands for survey data
- [XT] **xtnlogit** — Fixed-effects and random-effects multinomial logit models
- [U] **20 Estimation and postestimation commands**

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.

For suggested citations, see the FAQ on [citing Stata documentation](#).

